Chapter 10

Arrays
Приступая к изучению этой главы, Вы должны были уже познакомиться с тем, как используются переменные в Small Basic.
Давайте на мгновение вернемся к нашей первой программе, которую мы написали с использованием переменных:

TextWindow.Write("Enter your Name: ")

name = TextWindow.Read()

TextWindow.WriteLine("Hello " + name)

В этой программе мы спрашиваем у пользователя имя, и введенное имя присваиваем переменной name. Строчкой ниже мы выводим на экран надпись "Hello" с именем пользователя. Теперь, представим, что пользователей больше чем один, например 5. Как бы мы узнавали у них имена? К примеру, есть, такой способ сделать это:

TextWindow.Write("User1, enter name: ")

name1 = TextWindow.Read()

TextWindow.Write("User2, enter name: ")

name2 = TextWindow.Read()

TextWindow.Write("User3, enter name: ")

name3 = TextWindow.Read()

TextWindow.Write("User4, enter name: ")

name4 = TextWindow.Read()

TextWindow.Write("User5, enter name: ")

name5 = TextWindow.Read()

TextWindow.Write("Hello ")

TextWindow.Write(name1 + ", ")

TextWindow.Write(name2 + ", ")

TextWindow.Write(name3 + ", ")

TextWindow.Write(name4 + ", ")

TextWindow.WriteLine(name5)

Когда Вы запустите эту программу и выполните все действия, то получите на экране следующее:

Figure 49 – Не пользуетесь массивом?
Ясно видно, что есть лучший способ написания той же программы. Тем более, что компьютер действительно хорошо решает задачи с повтором, почему мы должны мучиться написанием одного и того же кода каждый раз для нового пользователя? Сложность заключается в том, что нужно получить и разместить больше чем одно имя пользователя, используя только одну переменную. Если мы сможем сделать это, тогда мы сможем использовать цикл For, который мы изучили ранее. В этом случае нам поможет работа с массивом.

Что же такое массив?

Массив это специальный класс переменной, которая может хранить больше чем одно значение за один раз. По существу это означает, - то, что вместо создания 5 переменных - name1, name2, name3, name4 и name5 для размещения пяти имен пользователей, мы можем создать одну - name и разместить в ней все имена. Таким образом, мы размещаем множество значений, при помощи "индекса". Например, имя можно разместить в любой из - name[1], name[2], name[3], name[4] и name[5]. Номера 1, 2, 3, 4 и 5 называются индексами для массива.

Несмотря на то, что названия name[1], name[2], name[3], name[4] и name[5] как бы говорят о том, что это разные переменные, в действительности это всё только одна. Вы можете спросить - «И в чём преимущество этого?». Лучшая черта массива та, что при сохранении значений, Вы можете определить индекс, используя другую переменную, которая позволит нам легко обращаться к массивам в циклах.

Теперь, давайте посмотрим на то, как мы можем поместить свое новое знание в массив, чтобы в дальнейшем его использовать, изменив нашу предыдущую программу с массивами.

For i = 1 To 5

  TextWindow.Write("User" + i + ", enter name: ")

  name[i] = TextWindow.Read()

EndFor

TextWindow.Write("Hello ")

For i = 1 To 5

  TextWindow.Write(name[i] + ", ")

EndFor
TextWindow.WriteLine("")

Намного понятнее при чтении, не так ли? Обратите внимание на две полужирные строки. Первая размещает значение в массиве, а вторая читает его из массива. Значение, которое Вы размещаете в name[1] не будет затронуто тем, что Вы размещаете в name[2]. Следовательно, в большинстве случаях Вы можете обработать name[1] и name[2] как две независимые переменные с одним и тем же отождествлением.

Figure 50 - Using arrays
Вышеупомянутая программа получает почти такой же результат как та, где массив не используется, за исключением запятой в конце имени Mantis. Мы можем это исправить, переписав цикл выводы следующим образом:

TextWindow.Write("Hello ")

For i = 1 To 5

  TextWindow.Write(name[i])

  If i < 5 Then

    TextWindow.Write(", ")

  EndIf

EndFor

TextWindow.WriteLine("")
Индексация массива
В нашей предыдущей программе Вы видели, что мы использовали только числа как индексы, чтобы разместить и получить значения от массива. Оказывается, что индексы не ограничены только целыми числами, и иногда более практично и полезно использовать также текстовые индексы. Например, в следующей программе, мы спрашиваем и размещаем различные части информации о пользователе и затем распечатываем информацию, которую просит вывести пользователь.

TextWindow.Write("Enter name: ")

user["name"] = TextWindow.Read()

TextWindow.Write("Enter age: ")

user["age"] = TextWindow.Read()

TextWindow.Write("Enter city: ")

user["city"] = TextWindow.Read()

TextWindow.Write("Enter zip: ")

user["zip"] = TextWindow.Read()

TextWindow.Write("What info do you want? ")

index = TextWindow.Read()

TextWindow.WriteLine(index + " = " + user[index])

Figure 51 – Использование не числовых индексов
Многомерные массивы
Скажем, Вы хотите создать массив, в котором разместить имя и номер телефона всех Ваших друзей, а затем чтоб было можно найти номер телефона всякий раз, когда Вам это необходимо – сделать своего рода телефонную книгу. Как бы мы подошли к написанию такой программы?

В этом случае, нам понадобится два набора индексов (также известные как размерность массива). Предположите, что мы определяем каждого друга по меткам его имени (начальные буквы). Это становится нашим первым индексом в массиве. Как только мы используем первый индекс, чтобы получить нашу удобную переменную, вторые из индексов, имени и номера телефона помогли бы нам получить данные для фактического имени и номера телефона того друга.

Способ, которым мы бы сохранили эти данные, походил бы на это:

friends["Rob"]["Name"] = "Robert"

friends["Rob"]["Phone"] = "555-6789"

friends["VJ"]["Name"] = "Vijaye"

friends["VJ"]["Phone"] = "555-4567"

friends["Ash"]["Name"] = "Ashley"

friends["Ash"]["Phone"] = "555-2345"

Так как у нас есть два индекса в одном и том же массиве, friends, то такой массив называют двумерным массивом.

Как только мы наберем эту программу, мы можем взять как данные для ввода прозвище друга и затем распечатать информацию, которую мы сохранили о нём. Вот полная программа, которая делает это:

friends["Rob"]["Name"] = "Robert"

friends["Rob"]["Phone"] = "555-6789"

friends["VJ"]["Name"] = "Vijaye"

friends["VJ"]["Phone"] = "555-4567"

friends["Ash"]["Name"] = "Ashley"

friends["Ash"]["Phone"] = "555-2345"

TextWindow.Write("Enter the nickname: ")

nickname = TextWindow.Read()

TextWindow.WriteLine("Name: " + friends[nickname]["Name"])

TextWindow.WriteLine("Phone: " + friends[nickname]["Phone"])

Figure 52 – Пример телефонной книги
Использование Массивов при построениях
Само по себе общее использование многомерных массивов должно представляться как работа с таблицей. У таблицы есть строки и столбцы, которые могут соответствовать двумерному, а также многомерному массиву. Простая программа, которая укладывает выходящие кубики в таблицу, показана ниже:

rows = 8

columns = 8

size = 40

For r = 1 To rows

  For c = 1 To columns

    GraphicsWindow.BrushColor = GraphicsWindow.GetRandomColor()

    boxes[r][c] = Shapes.AddRectangle(size, size)

    Shapes.Move(boxes[r][c], c * size, r * size)

  EndFor
EndFor
Эта программа добавляет кубики и устанавливает их таким образом, чтобы сформировать таблицу 8x8. В дополнение к укладыванию этих кубиков программа также размещает их в массив. Выполнение программы помогает нам следить за этими кубиками, а также использовать их в дальнейшем, когда нам это будет необходимо.

Figure 53 – Укладывание наших кубиков в таблицу
Например, добавление следующего кода к концу предыдущей программы сделало бы эти кубики анимированными к верхнему левому углу.

For r = 1 To rows

  For c = 1 To columns

    Shapes.Animate(boxes[r][c], 0, 0, 1000)

    Program.Delay(300)

  EndFor

EndFor

[image: image6.png]& SmallBasic Graphics Window.





Figure 54 - Наблюдение за построением таблицы
Индексы массива не чувствительны к регистру. Точно так же как обычные переменные соответствие индексов массива не должно соответствовать точному преобразованию букв в прописные.








